¿Qué es Option Explicit?

Option Explicit es una sentencia optativa pero recomendada que se ingresa al principio del código ASP de cada pagina. Las variables en ASP no necesitan ser declaradas para usarlas, pero es una buena costumbre hacerlo. Option Explicit sirve para que las variables deban ser declaradas si o si. 
Veamos como se usa. Es muy facil y solo sirve para obligar al programador (uds.) a declarar las variables. 
	<%
Option Explicit
'luego se declaran las variables sin problemas.
...
%>



Como ven es algo corto y simple. Sirve para acostumbrarse bien. Utilízenlo, que para algo existe.

Comentá tu código!

Muchas veces nos sentamos a escribir un script, en el cual ponemos toda nuestra atención y sabemos perfectamente que hace cada linea del código. Pero eso lo tenemos muy claro en el momento. ¿Que les quiero decir con esto? Que si otro día debemos retocar el código para que se adapte a nuestros nuevos objetivos, seguramente no recordaremos que significa cada variable, ni cómo trabaja nuestro programa. Otra cosa que se puede decir es que si pasamos nuestro script a otra persona, tendrá que perder valioso tiempo en entender como trabaja el código escrito
Toda esta introducción es para aconsejarles que siempre comentemos nuestro código.
Es simple, muy fácil y no perderemos tiempo al momento de entender los códigos viejos o que no conocemos.
Para comentar se utiliza el apóstrofe, ( ' ) en VBScript y la doble ( // ) en JavaScript. En HTML es asi: <!-- comentario -->. El servidor que lee nuestro script al encontrar un "'" deja de leer sabiendo que es un comentario que debe saltear.
Se puede comentar en cualquier lado. Arriba de la linea o al final. Comentemos algunas lineas. 

	<%
dim contar 'Definimos una variable.
'Hacemos un bucle en el cual la variable se incrementará hasta 10.
FOR contar = 1 to 10
'imprimimos el valor actual de la variable.
Response.Write (contar & " ")
Next
%>



Por cierto, este código hace lo siguiente... 

	1 2 3 4 5 6 7 8 9 10 



Pero notemos que un código comentado hace más fácil la lectura de él.
Comenten su código!! y ganarán tiempo y evitarán confusiones.

Recuerden tambien declarar el lenguaje que utilizarán para sus scripts al principio de todo.
   <% @ Language=VBScript % >
   <% @ Language=JScript %>

Declaración de las variables

En ASP como en casi todos los idiomas de programacion existen las variables. En ellas se pueden almacenar datos que pueden modificarse durante el transcurso del programa. Para declararlas usamos la instruccion Dim. No es obligatorio declarar una variable, pero es conveniente. Recuerden que si usamos Option Explicit si es obligatorio la declaración de la variable.
	<%
'con el apostrofe se escriben comentarios que no seran captados por el servidor.
'ahora explicaré el código.
'declaramos la variable nombre.
Dim nombre
'ahora se declaran dos variables en la misma linea, n1 y n2.
'las variables no son CaSe SensItivE, quiero decir que es lo mismo la variable AS que aS.
Dim n1, n2
%>




Asi de simple es la declaración de una variable. Con la sentencia Dim. Ahora asignaremos valores a esas variables y las modificaremos. Veamos como se hace. 

	<%
Dim nombre
Dim n1, n2
'ahora le asignamos valores a cada una de ellas.
nombre="CoRSA"
n1=10
'n1 vale 10.
n2=4
'n2 vale 4.
'distingan entre la variable nombre, la cual tiene su valor entre comillas, porque es.
'una cadena de caracteres, y n1 y n2, las cuales valen 10 y 4 respectivamente.
'Ahora modificamos n1.
n1 = n1 + n2
'n1 ahora vale 14. ¿Como es eso? Simple. n1 ahora es igual a n1 + n2 lo que es igual a n1 = 10 + 4.
n2 = n2 + 1
'n2 ahora vale 5 (4 + 1).
%>




Como ven es muy facil declarar y modificar variables. Ahora debemos aprender a utilizarlas para cosas útiles.
Primero que nada podemos imprimirlas por el explorador. Para eso necesitamos utilizar el Objeto Response.
En los siguientes capitulo se explicará precisamente este método.

Mi primer página ASP

Explicaremos a continuación una página ASP muy simple para que empiecen a familiarizarse con la sintaxis. La llamaremos prueba.asp. Describiremos cada una de sus lineas indicando que hace y cómo se usa.
	<%
'Utilizamos Option Explicit para no olvidar de declarar variables.
Option Explicit
'Declaramos nuestra variable nombre.
dim nombre
'le damos un valor. Éste está entre "" porque es una cadena de texto.
nombre="CoRSA"
'Declaramos otra variable, de nombre num1.
dim num1
'Le damos un valor. Ahora sin "" porque es numerica.
num1= 10
'Con esta sentencia la mostramos por pantalla.
Response.Write nombre
'Se pueden usar Tags de HTML en medio de ASP. Siempre entre "".
Response.Write "<BR>"
'Imprimimos la otra variable declarada.
Response.Write num1
%>



Esta página hace lo siguiente.
	CoRSA
10 



Ahora uds. deben decir... "tanto código para poner dos simples cosas??? Serían mas entendibles y más cortas en HTML".
Si, por supuesto que tienen razón. Pero este lenguaje se utiliza para cosas mas profundas como por ejemplo agilizar, resumir e interactuar con el usuario de la web. Esta última es la más importante. Pero antes de esas cosas, debemos aprender lo simple. Declarar variables, mostrarlas y usar HTML en medio de ASP. 

Para los que les parecio inutil y todavia no le ven utilidad, agregue lo que sería nuestra segunda página ASP. Aqui podran ver que en ASP se resume y se agiliza la programación. 

	Como harías esto?
Como harías esto?
Como harías esto?
Como harías esto?
Como harías esto?
Como harías esto?


	En HTML
	En ASP

	<FONT SIZE="1">Cómo harías esto?</FONT><BR>
<FONT SIZE="2">Cómo harías esto?</FONT><BR>
<FONT SIZE="3">Cómo harías esto?</FONT><BR>
<FONT SIZE="4">Cómo harías esto?</FONT><BR>
<FONT SIZE="5">Cómo harías esto?</FONT><BR>
<FONT SIZE="6">Cómo harías esto?</FONT><BR>
	<%var = 1
Do While var < 7
Response.Write "<FONT SIZE= "&var&">Cómo harías esto?<BR>"
var = var + 1
Loop%>



Tal vez esto no te sorprende, pero ASP es muy potente y estó es lo escencial.
Esto fue solo un ejemplo, que tal vez no se entienda. Solo intento mostrar la diferencia que existe.
Para entenderlo y escribirlo, puedes seguir leyendo otros articulos de esta misma web, donde se explicará.
Además el código en ASP es más variado, más dinámico y menos aburrido como lo podes observar

Tratamiento de Fechas

En ASP existen las funciones Date() y Time() las cuales piden la fecha y la hora del servidor donde son ejecutadas. Existen varios maneras de manejarlas a nuestro gusto para obtener exactamente lo que necesitamos. Podemos extraer el dia, el mes o el año. O podemos jugar con la función Now() para calcular cuanto tardo un usuario en presionar un enlace. Como veran no sólo se utilizan para imprimir el dia y la hora.
Veamos: 

	
<%
'Primero tomamos el instante y lo guardamos en la variable tiempo.
Dim tiempo
tiempo = Time()
Response.Write ("Esta es la hora que tiene mi servidor: " & tiempo)
fecha = Date()
Response.Write ("<BR> Hoy es el dia " & Day(fecha) & " del mes " & Month(fecha) & " del año " & Year(fecha))
%>


Veamos esto en accion!!!!!!!!

	Esta es la hora que tiene mi servidor: 8:21:27 PM
Hoy es el dia 17 del mes 3 del año 2002 




Tambien existen otras sentencias de manipulacion del tiempo.
	Sentencia Now() Dice día y hora.
Sentencia Time() Dice hora.
Sentencia Date() Dice día.
Sentencia Day(X) Extrae el número del dia (de 1 a 31).
Sentencia Month(X) Extrae el número del mes (de 1 a 12).
Sentencia Year(X) Extrae el número del año.
Sentencia WeekDay(X) Extrae el día de la semana en el que estamos (de 1 a 7).
Sentencia WeekDayName(X) Extrae el nombre del día.
Sentencia DateDiff(FormaDelDato, Dato1, Dato2) Saca la diferencia entre Dato1 y Dato2 y devuelve el valor con FormaDelValor

X debe ser una variable que contenga el Now() o el Date()
FormaDelDato es la manera en que calculara el DateDiff() 

yyyy - Año
q - Quincena
m - Mes
y - Dia del Año
d - Dia
w - Dia de la semana
ww - Semana del Año
h - Hora
n - Minuto
s - Segundo






Hagan pruebas con las diferentes sentencias para entenderlas mejor y apareceran nuevas ideas para su uso.

Objeto Response

El objeto Response se utiliza praa enviar información. Puede ser para mandar a imprimir texto o variables al explorador. Primero veamos la impresión de texto simple con el método Response.Write

Si ven en otros ejemplos ven una sintaxis similar a esta:
<%= "algo aqui" %> es lo mismo que escribir <% Response.Write "algo aqui"%>. 
O sea que el = reemplaza al Response.Write 

	<%
'con el apostrofe se escriben comentarios que no seran captados por el servidor.
'ahora explicaré el código.
Response.Write "Hola Usuario... bienvenido a nuestra Web <BR>"
Response.Write "Como ven se incluyen los tags de HTML en medio del código ASP"
'ahora imprmiremos una variable numerica.
dim numero
numero = 12
'ya declaramos la variable y le dimos un valor.
'ahora la sacaremos por pantalla.
Response.Write (numero & "<BR>")
'los parentesis no son necesarios aqui, solo hacen mas comprensible el código.
'el signo & es necesario para ingresar el TAG de HTML. Ver que los tags siempre van entre "".
'la modificamos y la imprimimos nuevamente.
numero = numero + 3
Response.Write numero
%> 



Estas lineas imprimen en el navegador lo siguiente: 

	Hola Usuario... bienvenido a nuestra Web
Como ven se incluyen los tags de HTML en medio del código ASP
12
15 



Otra caracteristica del metodo Response es la siguiente:
Response.Redirect redirecciona al usuario hacia otra URL.
Su funcionamiento es muy facil. Veamosló: 

	<%
'Esto enviaría al usuario al index.htm del sitio soloasp.com.ar
Response.Redirect "http://www.soloasp.com.ar/index.asp"
%>



Hay mas métodos para aplicar al objeto Response, por ejemplo Response.Cookies que lo veremos en el capítulo sobre Cookies
Tambien existen Response.Buffer, Response.Flush, Response.End, que los explicare en el capítulo Usando el buffer de la página
Objeto Request

Método Request.Form: se utiliza para pedirle (Request) a un formulario (Form), la información que fue ingresada en los diferentes campos. Luego los datos pueden ser almacenados en variables para trabajar más fácilmente con ellos.
Veamos un simple formulario. Puro HTML. Pero fijen la atencion en el METHOD y ACTION del FORM:
	Archivo: prueba.asp

<FORM METHOD="Post" ACTION="prueba2.asp">
Nombre: <INPUT NAME="nombre" SIZE="10">
<INPUT TYPE="Submit" VALUE="Enviar">
</FORM>



Este formulario pregunta el nombre de una persona. Al presionar el Boton de Enviar pasará la información a la página prueba2.asp, donde la recogeremos de la siguiente manera. Esto que sigue si es ASP 

	Archivo: prueba2.asp

<%
Dim nom
'declaramos nom y guardamos la info asi.
nom = Request.Form("nombre")
Response.Write nom
%>



Como ven no es algo complicado pasar datos por un formulario y recogerlos en otra página. Mas adelante podremos saber como guardar ese dato ingresado en una BD (Base de Datos) y almanear la información.
Como veras Request pide información enviada anteriormente.




Método Request.ServerVariables: se utiliza para pedirle a nuestro servidor algunas variables. Para que entiendan lo que quiero decir, podemos pedirle al servidor la direccion IP de un usuario, el nombre del dominio, la contraseña del cliente y muchas cosas mas.
La forma de utilizarlo es como veniamos haciendolo antes. De la siguiente manera:

	El nombre del servidor es: 
<B>
<%
Response.Write (Request.ServerVariables("SERVER_NAME"))
%>
</B> 



Esto pasaria como resultado lo siguiente:

	El nombre del servidor es: www.soloasp.com.ar


Lista de variables del Servidor
	Variable

	Descripción

	ALL_HTTP
	Todas las cabeceras HTTP enviadas por el Navegador.

	AUTH_PASSWORD
	La contraseña del cliente

	AUTH_TYPE
	El método de autentificación en los directorios de acceso restringido (El básico siempre)

	AUTH_USER
	Nombre de usuario autentificado en bruto (raw)

	CERT_COOKIE
	ID de usuario único para el certificado del cliente, devuelto como cadena de caracteres. Puede ser usado como firma para certificado al completo

	CERT_FLAGS
	· El bit0 es puesto a 1 si está presente el certificado del cliente. 

· El bit1 es puesto a 1 si la autoridad de certificados del certificado del cliente es inválida (no está en la lista de las reconocidas en el servidor) 

	CERT_ISSUER
	Issuer field of the client certificate (O=MS, OU=IAS, CN=user name, C=USA).

	CERT_KEYSIZE
	Número de bits de la clave SSL. Por ejemplo, 128

	CERT_SECRETKEYSIZE
	Numero de bits en el server certificador de  private key. por ejemplo, e.g. 1024.

	CERT_SERIALNUMBER
	Numero de serie deñ certificado de cliente.

	CERT_SERVER_ISSUER
	Issuer field of the server certificate.

	CERT_SERVER_SUBJECT
	Subject field of the server certificate.

	CERT_SUBJECT
	Subject field of the client certificate.

	CONTENT_LENGTH
	La longitud del contenido dada por el cliente

	CONTENT_TYPE
	El tipo de datos del contenido. Usado por las solicitudes que mandan infoermación como las sentencias de los formularios del HTML GET, POST, and PUT.

	GATEWAY_INTERFACE
	La revision del CGI usado por el servidor. El formato es CGI/revision.

	HTTP_<HeaderName>
	The value stored in the header HeaderName. Any header other than those listed in this table must be prefixed by HTTP_ in order for the ServerVariables collection to retrieve its value. 

Note The server interprets any underscore (_) characters in HeaderName as dashes in the actual header. For example if you specify HTTP_MY_HEADER, the server searches for a header sent as MY-HEADER.

	HTTPS
	Returns ON if the request came in through secure channel (SSL) or it returns OFF if the request is for a non-secure channel.

	HTTPS_KEYSIZE
	Number of bits in Secure Sockets Layer connection key size. For example, 128.

	HTTPS_SECRETKEYSIZE
	Number of bits in server certificate private key. For example, 1024.

	HTTPS_SERVER_ISSUER
	Issuer field of the server certificate.

	HTTPS_SERVER_SUBJECT    
	Subject field of the server certificate.

	INSTANCE_ID
	The ID for the IIS instance in textual format. If the instance ID is 1, it appears as a string. You can use this variable to retrieve the ID of the Web-server instance (in the metabase) to which the request belongs.

	INSTANCE_META_PATH
	The metabase path for the instance of IIS that responds to the request.

	LOCAL_ADDR
	Returns the Server Address on which the request came in. This is important on multi-homed machines where there can be multiple IP addresses bound to a machine and you want to find out which address the request used.

	LOGON_USER
	The Windows NT® account that the user is logged into.

	PATH_INFO
	Extra path information as given by the client. You can access scripts by using their virtual path and the PATH_INFO server variable. If this information comes from a URL, it is decoded by the server before it is passed to the CGI script.

	PATH_TRANSLATED
	A translated version of PATH_INFO that takes the path and performs any necessary virtual-to-physical mapping.

	QUERY_STRING
	Query information stored in the string following the question mark (?) in the HTTP request.

	REMOTE_ADDR
	La direccion IP del host remoto que hace la solicitud.  (hacer solicitud puede ser accesar la pagina simplemente)

	REMOTE_HOST
	El nombre del host que hace una sulicitud, si el servidor no tiene esta informacion, estapondra la de  REMOTE_ADDR y dejará la anterior vacia.

	REMOTE_USER
	Unmapped user-name string sent in by the User. This is the name that is really sent by the user as opposed to the ones that are modified by any authentication filter installed on the server.

	REQUEST_METHOD
	El metoso utlizado para mandar los datos del formulario HTML, estos pueden ser  GET, HEAD, POST.

	SCRIPT_NAME
	La ruta virtual donde el script es ejecutado. se usa en las referencias a si mismo en las URLs

	SERVER_NAME
	Devuelve el Nombre de Host del servidor, el alias dl DNS ola direccion IP  

	SERVER_PORT
	The port number to which the request was sent.

	SERVER_PORT_SECURE
	A string that contains either 0 or 1. If the request is being handled on the secure port, then this will be 1. Otherwise, it will be 0.

	SERVER_PROTOCOL
	El nombre y version del protocolo que se usa para las requests. EL formato es protocol/revision.

	SERVER_SOFTWARE
	Regresa el nombre y version del del software servidor que reponde las "requests" y corre la CGI's. El formatro es name/version.

	URL
	Devuelve la base de donde parte una URL


Condicional IF, y bucle Do While... Loop

Los bucles IF se utilizan constantemente en cualquier tipo de programación. Sirven para comprobar si una variable cumple con un requisito o esta dentro de algun rango. Simplemente se fijan en el valor de la variable y si este concuerda con nuestra ordenes del IF, entonces continua. Si la variable esta fuera de lo que le pedimos... saltea el IF. Para su mejor comprención veamos un facil ejemplo: 

	<%
dim mes, fecha
'ya declaramos las variable y ahora le damos un valor.
'la variable fecha guardara la fecha completa.
fecha = Date()
'y la variable mes, extraerá el mes de la variable con la fecha.
mes = Month(fecha)
'Imprimimos por pantalla el numero del mes.
Response.Write (mes & "<BR>")
'Y luego comprobamos si mes es igual mayor o menor que 5.
IF mes = 5 Then
Response.Write "Estamos en el mes de Mayo"
ElseIf mes < 5 Then
Response.Write "Todavia falta para Mayo"
Else
Response.Write "El mes de Mayo ya paso"
End If
...
%> 



Observar el código:
Notar que se uso IF, ElseIF y Else. El comando IF pregunta por algo determinado. Y el comando Else abarca todo lo que NO abarco el IF
Es por eso que para decir "El mes de Mayo ya paso" usamos Else, ya que si mes no es igual a 5, ni es menor... tiene que ser mayor.
All terminar el bucle IF no olvidar nunca End IF, ya que tendremos un error al hacerlo. Ahora veremos el programa escrito arriba en acción. fijense que la variable mes se modifica automaticamente todos los meses, indicando el numero del mes en el que nos encontramos.

	3
Todavia falta para Mayo 




Los Bucles Do While... Loop funcionan de manera similar aunque hay diferencias. Do While significa hacer mientras... o sea que mientras se este cumpliendo una condición continuará con el bucle. Cuando deje de cumplirse, saltará el bucle.
Un ejemplo.

	<%
conta = 1
'Ahora... mientras que conta sea menor que 6... hacer.
Do while conta < 6
Response.Write ("<FONT SIZE=" & conta & "> Me llamo Corsa </FONT><BR>")
conta = conta+1 
'como ven utilizamos la variable conta para dos cosas.
'la primera: imprimir "Me llamo CoRSA" 5 veces. (girar por el bucle)
'la segunda: utilizar la misma variable para cambiar el SIZE de la FONT.
Loop
%>



Tener cuidado con la sintaxis de HTML dentro del código ASP.
Este ejemplo producira lo siguiente!

	Me llamo Corsa 
Me llamo Corsa 
Me llamo Corsa 
Me llamo Corsa 
Me llamo Corsa 


Bucle For... Next - [Por CoRSA]

Otra manera de crear un bucle (repetición), es con la sentencia For... Next. Con esto podemos crear un bucle repitiendo varias operaciones. Para hacer esto, se elige una variable que acturá como contador de las vueltas del bucle. Cuando alcanze un valor determinado, finalizará el bucle. Primero se iguala la variable que cuenta las vueltas al número que deseamos comenzar (For i=0). Por defecto, la variable que cuenta las vueltas del bucle se incrementa en 1, pero se puede asiganr otro valor de incremento. Este valor se pone despues de la palabra Step. 
Escribí un ejemplo de bucle For... Next, donde me calcula el cuadrado de la variable que está contando las vueltas. Es así: 
	<%
'Definimos la variable contador.
dim cuen
'Le decimos al bucle que la variable cuen empieza en 0, y que debe finalizar el bucle cuando sea igual a 10.
'Con Step indicamos que se debe sumar solo 1 por ciclo.
For cuen = 0 To 10 Step 1
'Mandamos un mensaje, que calcula el cuadrado de la variable.
response.write cuen*cuen &" es el cuadrado de "& cuen &"<BR>"
'Ponemos un condicional IF, que cuando se cumpla muestre un mensaje.
if (cuen=10) then 
response.write "Ha terminado el bucle..."
end if 
'Volvemos al For, siempre y cuando la variable cuen no haya llegado a 10.
Next
%>




A continuación se muestra el bucle trabajando. Notar que el Step es de 1... se puede elegir cualquier incremento. 
	0 es el cuadrado de 0
1 es el cuadrado de 1
4 es el cuadrado de 2
9 es el cuadrado de 3
16 es el cuadrado de 4
25 es el cuadrado de 5
36 es el cuadrado de 6
49 es el cuadrado de 7
64 es el cuadrado de 8
81 es el cuadrado de 9
100 es el cuadrado de 10
Ha terminado el bucle... 


	
Include - [Por CoRSA]

Con Include podemos llamar a otro archivo e insertarlo en el código donde estamos trabajando. Esto es muy util cuando tenemos que repetir en todas las páginas lo mismo. ¿Qué quiero decir? Simple. Uds crean y diseñan, por ejemplo, un menu el cual debe aparecer en todas las páginas de su sitio. Lo crean en un archivo que se llama menu.asp, y luego desde cada página llaman a ese archivo e insertan su contenido en donde lo necesitamos. Mostremos un ejemplo para ir entendiendo como funciona, y luego seguimos con la explicación.

Creamos el archivo menu.asp con el menu. En nuestro ejemplo un "menu" horizontal para el comienzo de una página cualquiera. 

Archivo: general.asp

<%
'Abrimos un Sub.
Sub menu
%>
<TABLE BORDER="0" ALIGN="center" WIDTH="90%">
<TR>
<TD>Home Page</TD>
<TD>Ejemplos</TD>
<TD>Historia>/TD>
<TD>Contactate</TD>
</TR>
</TABLE>
<%
'Cerramos el Sub.
End Sub
%>

Archivo: homepage.asp

<!-- #include file="general.asp" -->
<%
'Llamamos al Sub.
Call Menu
%>
Resto de la página general.asp 

Archivo: ejemplos.asp

<!-- #include file="general.asp" -->
<%
'Llamamos al Sub.
Call Menu
%>
Resto de la página de ejemplos. 


¿Qué logramos con esto?
La cosa más importante es que cuando modificamos el archivo general.asp, cambiaran todas las páginas. Es lógico, ya que todas llaman a ese archivo. Entonces cuando necesitemos cambiar algo en el menu, sólo tenemos que cambiar una página y no todas.

¿Cómo se hace para incluir un archivo en otro?
Bueno, viendo el ejemplo podemos saberlo, pero lo explicaré para que se entienda mejor, y aclarar algunas cosas. La primera es la forma de llamar al archivo a incluir. Disponemos de dos métodos: #include file o #include virtual.
Con #Include File necesitamos que estén en el mismo directorio el archivo a incluir (menu.asp) y el archivo que lo pide (homepage.asp o ejemplos.asp)
Con #Include virtual esto no es necesario. Pero debemos decir el path (camino) donde está el archivo a incluir. Por ejemplo: <!-- #Include Virtual="inclu/general.asp" -->
Esto buscaria siempre el archivo general.asp en el directorio inclu/

Sub y Call:
Habrán visto que en el archivo general.asp se abría un Sub llamado menu y luego se cerraba. Podemos crear varios Sub en el mismo archivo y luego llamar a cada uno por separado con la instrucción Call. Después de Call se debe poner el nombre de la Sub. Fijense bien como se usa y ahorrarán lineas y lineas de código. 

Algo muy importante:
Si no usamos los Subs, también se puede llamar al archivo directamente con la línea de #Include Virtual, pero no se podría usar código ASP en el archivo llamado. Lo que quiero decir es que si no usamos Subs, no pueden escribir ASP, en cambio si creamos un Sub si podremos. 


	
Cookies

¿Quién no ha oido hablar sobre las cookies? ¿Sabes que tienes un directorio en tu rigido que tiene todas las cookies que te crean las páginas? Bueno, supongo que escuchaste de ellas... pero tal vez no sabes qué son, ni usarlas. No te preocupes, para eso estas aquí.
Primero les dire que son las Cookies. Una cookie es un archivo txt que se graba usualmente en el directorio \Windows\Cookies/. Si quieren hechar un vistazo, haganlo ahora mismo, y veras que tienen unos archivos en ese directorio. Pero las cookies (archivos en tu rígido), guardan información, que el servidor pide al usuario al cargar determinada página. La información que guardan es variada. Cada sitio puede crear cookies con informacion diferente. Por ejemplo pueden guardar tu nombre, o las veces que ingresaste a la página. Veamos como hacer una cookie, luego como pedirla al usuario y por último imprimirla.
Veamos un ejemplo simple que sirva para identificar a cada usuario.
<%
'Primero que nada nos fijamos si ya lleno el formulario.
If Request.Form ("nombre") <> "" then
'Si ya se completó, creamos la Cookie.
'Recuerden que la función Trim es para sacar los espacios.
'en blanco a izquierda y derecha si existen.
Response.Cookies ("nombre") = Trim(Request.Form ("nombre"))
'Con la siguiente sentencia se indica la "vida" de la Cookie.
Response.Cookies("Nombre").Expires = "1/1/2000"
End If
%>
'Empieza la página.
<HTML>
<BODY>
<%
'Nos fijamos si ya existe una Cookie.
If Request.Cookies ("nombre") <> "" Then
'Si existe, la mostramos asi.
Response.Write "La cookie se ha creado. Hola " & Request.Cookies ("nombre")
Response.Write "<BR>Como vez, esto de las cookies es util"
Else
'Mostramos el formulario para que sea completado.
Response.Write "Ingresa tu nombre en el formulario. Crearemos una cookie con él."
%>
<FORM ACTION="mismapagina.asp" METHOD="POST">
<INPUT NAME="nombre" SIZE="10">
<INPUT TYPE="Submit" VALUE="Crear Cookie!">
</FORM>
<% End If %>


Notar que la ACTION del FORM envia los datos a la misma página. Es por eso que antes de los tags de HTML se chequea si fue completado el form y se crea la cookie.
El mismo código trabajando esta aqui abajo. Para probar las Cookies llenen el campo: 

Ingresa tu nombre en el formulario. Crearemos una cookie con él. 

Principio del formulario

Final del formulario




Objeto Session

El objeto Session se utiliza en ASP para abrir una sesion, que tiene por defecto un tiempo de 20 minutos. Durante esta sesión un usuario , por ejemplo, puede estar logueado en un sitio y permanecer hasta que pase el tiempo de la sesión. Y si no estubo activo durante X minutos, deberá loguearse nuevamente.
Pero este no es su unico uso. Por supuesto que se lo puede utilizar para muchas cosas. Yo en mi sitio, lo uso para que un usuario no pueda votar en la encuesta mas de 1 vez por hora. ¿Cómo hago esto? Simple. Cuando detecto que un usuario realiza la votación, creo uns sesión que dura 1 hora, y siempre que intente votar dentro de ese lapso de tiempo, se lo impediré chequeando la variable de sesión. Si esta en valor True significa que no han trancurrido aún los 60 minutos.
Espero no haberlos maredo. Igual recueden que cualquier duda, me la pueden enviar por mail o llenando un formulario en la sección Contactanos.

Bueno, veamos un pequeño ejemplo para ver su sintaxis y funcionamiento.
	<%
'primero que nada, me fijo si la Session esta en True.
If Session("yaestube") = True then
'si es True, mando este texto.
Response. Write "Como vez, el texto es diferente, ya que la <B>Session</B> esta activa."
'si NO es True, mando esto.
Else
Response.Write "Hola... la variable de Session no existía, pero se acaba de crear"
'Aqui creo la session. Asi la proxima vez se cumple el IF y se muestra el texto anterior.
'Esta es la sintáxis para crear una Session.
Session("yaestube") = True
End If
%>



Aca esta el código en acción: (hacer refresh para ver el cambio)
	Hola... la variable de Session no existía, pero se acaba de crear 



Para cerrarle una sesión al usuario se utiliza el metodo Abandon. La sitáxis es: <% Session.Abandon %>
Con esto se finaliza la sesión aunque quede tiempo restante.
Para asignar el tiempo de duración de una sesion usamos esa linea: <% Session.TimeOut = 8 %> 
Es obvio que el tiempo para cerrar la sesion en nuestro ejemplo es de 8 minutos.
Objeto Server

El objeto server es el que utilizamos cuando hacemos la Conexion a una BD o cuando creamos un RecordSet. Pero además, tiene otros métodos y propiedades que podemos usar al programar nuestra página ASP. A continuación veremos algunos de esos métodos y los describiremos para que conoscan como funcionan y que acción tienen.

Método CreateObject("Id"): Sirve para crear objetos en nuestro servidor. Objectos que sirven para mandar mails ("CDONS.NewMail"), para conectar a base de datos ("ADODB.Connection") y para obtener registros de las BD ("ADODB.RecordSet"). Se vera su uso en capitulos posteriore. Los referidos a Base de Datos.

Método HTMLEncode("cadena"): Lo podemos utilizar para solucionar el problema con los caracteres especiales. Reemplaza "esos" caracteres por los de HTML. Reemplazaría el signo ">" por "&gt;". Para entenderlo miremos un ejemplo: 

	<% 
dim cadena
cadena = Server.HTMLEncode("hola ""amigo"" como verás no hay problema si usas comillas o <B>tags de HTML</B>")
Response.Write cadena 
%>



Esto devolveria el siguiente texto:

	hola "amigo" como verás no hay problema si usas comillas o <B>tags de HTML</B> 



Pero si miran el código HTML de está pagina verán lo siguiente, ya que el HTMLEnconde transformó los caracteres: 

	hola &quot;amigo&quot; como ver&#225;s no hay problema si usas comillas o &lt;B&gt;tags de HTML&lt;/B&gt; 




Método Execute("URL"): Solo con el IIS 5.0 se puede utilizar este método, por el cual podriamos correr otra página desde la que estamos. La página elegida se ejecuta dentro de la nuestra, y al finalizar la segunda página, se continúa con el resto de la nuestra. Usamos <% Server.Execute("http://www.algo.com") %>

Método MapPath("URL"): Devuelve la ruta donde nos encontramos. Usamos <% Server.MapPath("archivo.asp") %> 

Una propiedad de el Objeto Server es ScriptTimeout. Esta propiedad es la que indica el tiempo que debe trancurrir para que se deje de cargar un script. El tiempo por defecto es de 90. Nosotros lo podemos modificar si tenemos que cargar un script demasiado largo que necesita de mas tiempo.
Para usarlo es simple: <% Server.ScriptTimeOut = 90 %>

Usando el Buffer de la página

Las páginas ASP se procesan en el servidor web, quien devuelve al cliente puro código HTML. Este procesamiento se contruye en el Buffer. Si estamos usando el IIS 5 (Internet Information Server 5) que viene en Windows 2000, el método Buffer del objeto Response viene por defecto en valor True. Quiero decir que viene Response.Buffer = True. En el IIS 4 o PWS esta en False. Pero la podemos poner a True cuando queramos.
Se estarán preguntando que es lo que cambia... Bueno, si el Buffer esta en True, el servidor procesará toda la pagina ASP, creará la página para el cliente en HTML y cuando finalize todo ese trabajo, enviará todo junto. Con el Buffer en False, el servidor va enviando la información que ya proceso, mientras continua trabajando con el resto.
Seguramente notaron que en algunos sitios, nos baja toda la información de un solo golpe (texto, imagenes, sonido), en cambio lo más comun hoy en día es que baje a medida que se carga. Esto se puede manejar a nuestro gusto.
Veamos a continuación como podemos trabajar con el Buffer y como se usa correctamente:

Propiedad Buffer: ya lo he explicado en la introducción. Puede estar True o False. Si está en True el servidor procesa todo y luego lo envia. Si está en False va enviando a medida que procesa.

Método Flush: Se puede utilizar solamente cuando el Buffer está en True. ¿Para que sirve? Dijimos que cuando el Buffer está en true, toda la información se envia cuando es procesada completamente. Con Response.Flush pediremos todo lo que está procesado hasta ese momento.
Es útil para usarlo después de fragentos "pesados" (imagenes, menues), porque de esta manera será enviado el menu y luego seguirá con el resto de la página.

	<%
Response.Buffer = True 
'parte pesada de la página...
....
Response.Flush
'Continuamos con la página. 
...
%> 



Método Clear: Se utiliza para borrar todo el Buffer. Su sintaxis es simple <% Response.Clear %>

Método End: Se utiliza para terminar la ejecución de un Script. Su uso más comun es para detener cualquier página si detectamos que el usuario no está más. Eso lo hacemos de este modo:

	<%
'parte pesada de la página.
...
'comprobamos si el cliente sigue activo.
if not Response.IsClientConnected then 
Response.End 
end if
%>

	Evitando la Caché - [Por CoRSA]

El navegador guarda una copia de las páginas visitadas en la caché. Cuando retorna al mismo sitio lo que hace es verificar si ya visitó esa página y cuál es la fecha del archivo. Si la fecha de modificación concuerda con la del servidor, la página se carga desde la caché. Si no, baja del server. Veremos a continuacion como evitar eso, y que siempre baje el archivo desde el servidor, evitando la caché del navegador.

Método Response.Expires: Con este método definimos los minutos que se usará la caché. Si ponemos 3 significa que a los 3 minutos dejará de usar la caché, y pedirá la página al servidor. Obviamente si ponemos 0, se evita totalmente la caché, ya que pedirá directamente la página al server. Algo importante es que debemos ubicarlo antes de los encabezados HTTP. Su forma de usarlo es el siguiente.

<% Response.Expires = 0 %>


Método Response.ExpiresAbsolute: Se utiliza si queremos que cierta página caduque un día determinado. A partir de ese día no se cargará desde la caché sino desde el servidor directamente.

<% Response.ExpiresAbsolute = #4/12/2000 10:00:00# %>


Método Response.AddHeader: Tiene varias funciones, pero sólo comentaremos la siguiente que sirve para no llegar a guardar la caché de la página en cuestión.

<% Response.AddHeader "PRAGMA", "NO-CACHE" %>




Tipos de conexión a una BD (Base de Datos)

1 - Conexión con DSN (ejemplo para Access)

La conexion con DSN es la más cómoda, pero sólo se puede utilizar si tenemos acceso al Panel de Control de la máquina servidor. Por supuesto si estamos contruyendo una intranet tenemos el server a nuestro alcanze y a su Panel de Control.
Si simplemente estamos aprendiendo ASP y usamos el PWS (Personal Web Server) o el IIS 4 de NT tambien disponemos de esta comodidad.
Veamos como se realiza la conexion a una base de datos de Microsoft Access:

Creamos nuestra base de Datos en Microsoft Access y la guardamos. Luego vamos a Inicio > Configuracion > Panel de Control y alli elegimos Fuentes de Datos ODBC
Al ingresar nos encontramos con una pantalla que es el administrador de origenes de datos ODBC. En la solapa DSN de Usuario presionamos el boton Agregar. Luego seleccionamos Microsoft Access Driver (*.mdb) y presionamos Finalizar. Ahora se hara la conexión ODBC. Presionamos el boton Seleccionar y elegimos nuestra Base de Datos e ingresamos el nombre de la base en el primer campo.
Por último el boton Aceptar
Si todo salio bien debería aparecer el nombre de nuestra Base de Datos en la solapa DSN de usuario y ya tendremos hecha nuestra conexion ODBC a BS.

Ahora debemos conectar la base de datos en la pagina ASP 
	<%
'Definimos la variable para la conexión.
Dim Conex
Set Conex = Server.CreateObject ("ADODB.Connection")
'y ya estamos conectados a nuestra base de datos.
Conex.Open "nombre de la BD"
'aqui abrimos la tabla. ...
%>




Como vieron no es algo dificil... solo hay que aprenderser la linea de conexión. Pero recuerden que esto no servira si suben su BS y su sitio a internet. Para esto deben usar la siguiente conexón. 

2 - Conexión sin DSN 

Este tipo de conexión es más complicada, pero es lo que debemos utilizar si queremos olvidar el panel de control, ya que hacemos la conexión a la base de datos mediante comandos. Ademas es mas rápida ;)
Observar bien bien la sintaxis... ya que es lo que nos trae los problemas siempre.

Para Access usando ODBC: 

	<%
Dim Conex
'Creamos el objeto de conexion ahora...
Set Conex = Server.CreateObject ("ADODB.Connection")
Conex.open "DRIVER={Microsoft Access Driver (*.mdb)}; DBQ=C:\misitio\db\mibase.mdb;"
%>



Para Access 97 usando OLEDB: 

	<%
Dim Conex
'Creamos el objeto de conexion ahora...
Set Conex = Server.CreateObject ("ADODB.Connection")
Conex.Open "Provider=Microsoft.Jet.OLEDB.3.51; Data Source=C:\misitio\db\mibase.mdb;" 
%>



Para Access 2000 usando OLEDB: 

	<%
Dim Conex
'Creamos el objeto de conexion ahora...
Set Conex = Server.CreateObject ("ADODB.Connection")
Conex.open "Provider=Microsoft.Jet.OLEDB.4.0; Data Source=C:\misitio\db\mibase.mdb;"
%>



Para SQL Sever: 

	<%
Dim Conex
'Creamos el objeto de conexion ahora...
Set Conex = Server.CreateObject ("ADODB.Connection")
Conex.Open "driver={SQL Server};server=TU_SERVIDOR; database=NOMBRE_BASE; uid=sa;pwd=" 
%>


Importante: En los 3 primeros ejemplos, referidos todos a conexiones con bases de datos Access, use un path fijo. Lo puse asi, para el mejor entendimiento, pero generalmente se hace uso de la funcion Server.MapPath() la cual devuelve el path donde ejecutamos el script

Les intentaré explicar el uso de esa función mediante un ejemplo que no tiene que ver con conexion a base de datos, pero intentaré que se relacione.

	<%
Dim path
'guardamos en la variable path lo que devuelve la función
path = Server.MapPath("./")
Response.Write path
%>



El código ejecutado, muestra el contenido de la variable path 

	C:\InetPub\WWWRoot\directorio_actual 


Como ven la variable path contiene el directorio donde se ejecuta el script.
Entonces esto se podría usar para indicar el path a la base de datos (que en los ejemplo anteriores lo escribiamos a mano)

Si quedan dudas sobre esto, ya que es un tema que comunmente trae confusiones, pueden preguntar en el Foro de discusion
	
Mostrar tablas de Access

Obviamente antes de mostrar una información en una tabla debemos conectarnos a ella. En el artículo anterior, como conectarnos a una BD, vimos como se hacía la conexión. Ahora, después de conectarnos a la BD, abriremos una tabla de la BD y mostraremos su información por la pantalla.
Veamos como mostrar nuestros datos!... 
<%
Dim oConn, RS
'conectamos a la tabla.
Set oConn = Server.CreateObject ("ADODB.Connection")
oConn.Open "DRIVER={Microsoft Access Driver (*.mdb)}; DBQ=D:\Webs\Soloasp\db\mibd.mdb"
'Abrmios el RecordSet (accedemos a los datos de la BD).
Set RS = Server.CreateObject ("ADODB.RecordSet")
'entre " " va el nombre de la tabla, y luego de la "," se indica la BD que contiene esa tabla.
'El número que le sigue, el "1", es la forma de abrir la tabla.
RS.Open "contador", oConn, 1
'Mostramos los campos hasta que nuestro RS no termine.
'Cuando termine... sale del bulcle automaticamente.
Do While Not RS.EOF
Response.Write RS("hits") &"<BR>"
Response.Write RS("fecha")
'MoveNext nos movera al siguiente registro de la tabla. A la siguiete fila.
RS.MoveNext
Loop
'primero cerramos los objetos y luego los limpiamos.
RS.Close
oConn.Close
Set RS = nothing
Set oConn = nothing
%>


En resumen. con RS.Open "datos" estamos abriendo la tabla.
Con Do While Not RS.EOF... Loop, indicamos que se cumpla el bucle hasta que llegue al último registro. Justamente esto significa EOF. Cuando EOF es true, significa que estamos parados en la ultima posición de la tabla.
Con Response.Write RS("campo"), mostramos por pantalla el "campo"
RS.MoveNext nos pasa a la siguiente linea o fila.
Y por último nunca se olviden de cerra el RecordSet y la conexión a la BD. Y limpiarlos.


¿Que es SQL? Funciones Básicas

SQL es un lenguaje universal dedicado a las bases de datos. Es un lenguaje de sintaxis simple y muy potente. Mediante él se puede recorrer, modificar o borrar registros de las tablas de datos. 
A continuación escribi una guía básica para aprender SQL. Las funciones principales y el modo de utilización. Veámoslo ahora:
	Nota: El asterisco (*) es un comodin, que selecciona todas las columnas dentro de los límites que pongamos.

Propiedad SELECT: Hace la selección en una tabla de la BD.

SELECT * FROM datos
'Esta sentencia seleccionaria todos, absolutamente todos los registros dentro de la tabla datos.

SELECT * FROM datos WHERE usuario='juan'
'Seleccionariamos todos los registros dentro de la tabla datos que tengan como usuario a "Juan".

Propiedad INSERT INTO: Agrega un nuevo registro a la tabla elegida

INSERT INTO datos (usuario) VALUES ('corsa')
'Insertamos en la tabla datos en la columna usuario, un registro nuevo.

INSERT INTO datos (usuario, edad) VALUES ('corsa',20)
'Aqui insertamos 2 registros al mismo tiempo.

Propiedad DELETE: Borra registros de nuestra tabla

DELETE FROM datos WHERE usuario = 'pedrito'
'Borramos los regitros donde el usuario sea "pedrito".

DELETE FROM datos WHERE usuario = 'Mario' AND edad = 16
'Borramos solo los usuario de nombre Mario que tenian 16 años.

Propiedad UPDATE: Actuailiza registros, modificando datos ya existentes.

UPDATE datos SET usuario = 'pedrito'
'Esta modificación renombrará todos los usuaarios a "pedrito".

UPDATE datos SET usuario = 'Mario' WHERE edad = 16
'Modificamos solo los registros que tenian 16 años. Ahora todos los usuario de 16 años se llaman "Mario".

Modo de usar SQL en nuestras bases de datos:

Podemos guardar la sentencia SQL en una variable y llamarla para usarla:

<%
Dim sql, oConn, RS
sql = "SELECT * FROM datos"
'Conectamos a la BD.
Set oConn = Server.CreateObject("ADODB.Connection")
Set RS = Server.CreateObject("ADODB.RecordSet")
oConn.Open "DRIVER={Microsoft Access Driver (*.mdb)}; DBQ=D:\Webs\Soloasp\db\mibd.mdb"
'ahora abrimos nuetro RecordSet con los parámetros del sql que pusimos antes.
RS.Open sql, oConn, 1
...
'Este DO WHILE imprimiría todos los registros en la columna nombre.
Do While not RS.EOF
Response.Write RS.Fields("Nombre")
RS.MoveNext
Loop
...
'Cerramos y limpiamos todo.
RS.Close
oConn.Close
Set RS = Nothing
Set oConn = Nothing
%>




Procedimientos y Funciones en VBScript - [Por CoRSA]

Las funciones y procedimientos se utilizan para ordenar, organizar y hacer un mejor codigo.
Primero les hablaré de los procedimientos, ya que son mas sensillos que las funciones.

Procedimientos:
Los procedimientos contienen una serie de comandos o instrucciones en VBScript, encerrados entre las instrucciones Sub y End Sub.
Estas instrucciones dentro del procediemintos, realizan acciones, pero a (diferencia de las funciones) no devuelven ningun valor.
Los procedimientos pueden aceptar argumentos. Los argumentos son variables que se pueden enviar al procedimiento y utilizarlas para realizar las acciones.
Veamos un ejemplo corto de procedimientos sin pasar argumentos

	<%
'Iniciamos un procedimiento...
Sub algo()
Response.Write "Hola en este momentos son las: "& Time()&"<BR>"
End Sub
%>

<%
'llamamos al procedimiento por su nombre... 
algo
algo
algo
%>



El codigo este ejecutado produciria...

	Hola en este momentos son las: 8:33:13 PM
Hola en este momentos son las: 8:33:13 PM
Hola en este momentos son las: 8:33:13 PM



Como ven llamamos 3 veces seguidas al procedimiento algo, entonces se ejecuto 3 veces... Solo se ejecuta el extracto de código encerrado entre Sub algo() y End Sub.

Dijimos que ese procedimiento no tenia argunmentos. Eso se declara en la definicion del procedimiento.
Un procedimiento sin argumentos se declara asi: 
    Sub nombre_procedimiento ()

En cambio uno con argumentos se declara con ciertas variables dentro de los paréntesis asi: 
     Sub nombre_procedimiento (argumento1, argumento2)

Ahora veamos un procedimiento con un solo argumento. Un ejemplo sensillo.

	<%
'Comienzo un procedimiento, que tiene un argumento 
'a lo que se meta en el argumento, lo pongo en letras mayusculas con la funcion UCase() 
'y lo imprimo en negrita... 
Sub negrita_mayuscula(argumento1)
argumento1 = UCase(argumento1)
Response.Write "<B>"&argumento1&"</B>"
End Sub
%>

<%
'declaro una variable cualquiera... 
nombre = "Corsa"
'llamo al procedimiento, y le paso la variable como argumento. 
negrita_mayuscula(nombre)
%>



El codigo este ejecutado produciria...

	CORSA 


Notar que la variable nombre era una cadena de caracteres asi: "Corsa", sin negrita y con una sola mayúscula. Luego de pasarla por el procedimiento, salio pintada de negrita y toda en letras mayusculas.
El funcionamiento de los argumentos es mas o menos asi... Uds notan en el ejemplo que la variable nombre pasa al procedimiento como argumento... ya que va entre paréntesis. Esa variable toma el nombre (en nuestro ejemplo) de argumento1
A esta nueva variable argumento1 que contiene la informacion que tenia nombre en ese momento, se le hacen las acciones.

Algo muy importante: En los procedimientos no estamos modificando las variables que pasan como argumentos reales en ningun momento (en nuestro ejemplo la variable a la cual llamo real es nombre). Si hacemos un Response.Write de la variable nombre veremos que se imprime tal cual como fue definida.

	
Procedimientos y Funciones en VBScript - [Por CoRSA] 

Ahora veremos como usar funciones y sus diferencias con los procedimientos.

Las funciones se escriben entre Function y End Function. Al igual que los procedimientos, aceptan argumentos y tambien se encierran entre paréntesis al declarar la función.

La importante diferencia entre los procedimientos y las funciones, es que estas últimas, devuelven un valor. Veremos a continuacion en ejemplo mas simple de funcion y veremos a lo que me refiero.

<%
'definimos la funcion con un argumento
Function may(argumento1)
   argumento1 = UCase(argumento1)
   'la siguiente linea produce la devolucion del valor...
   may = argumento1
End Function

nombre = "lala"
response.Write may(nombre)
%> 


El código ejecutado produce... 

LALA 

Bueno, como verán hay algunas diferencias entre las funciones y los procedimientos, aunque pueden parecer similares.
Pero noten que en la linea may = argumento1 se está diciendo que la funcion devolverá el contenido de argumento1. Luego, fuera de la función se llama a la funcion por su nombre pasando el argumento correspondiente...

Como las funciones devuelven valores, uno puede guardarlos, en cambio los procedimientos no devuelven nada. Recordar que para mostrar el procedimento que se hacían, se ponía un response.Write dentro del Sub

Hay mucho por hablar sobre las funciones... por ejemplo son usadas generalmente para agrupar mini (o no tan mini) procesos que deben afectar a ciertas variables. Entonces se usan para organizar el código y no tener problemas a la hora de entender o depurar errores...
Se podrían decir mas cosas... pero para una introduccion creo que está bien.

	
Login con usuario y contraseña desde una BD de Access - [por CoRSA]

El ejemplo a continuación sirve para chequear si el nombre de usuario ingresado en un formulario, coincide con su contraseña. Si no es asi, denegaremos el ingreso a la siguiente página.
Veamos el ejemplo que sigue.

Este codigo de Puro HTML va en la primer página:
Archivo: pagina1.asp

<FORM ACTION="pagina2.asp" METHOD="post"><BR>
<B>Usuario:</B> <INPUT NAME="usuario" SIZE="12"><BR>
<B>Contraseña:</B> <INPUT NAME="password" SIZE="12"><BR><BR>
<INPUT TYPE="Submit" VALUE="Ingresar">
</FORM> 

Este código va en la página a donde se envian los datos del FORM. En mi ejemplo seria pagina2.asp.

Archivo: pagina2.asp

<%
'Primero nos fijamos si ambos campos fueron completados.
'Si no se cumple, redireccionamos a pagina1.asp
If Request.Form("usuario") = "" OR Request.Form("password") = "" then
Response.Redirect "pagina1.asp"
Else
Dim usuario, password, oConn, RS
'Guardamos los datos del Form en variables.
usuario = Trim(Request.Form("usuario"))
password = Trim(Request.Form("password"))
'Conectamos a nuestra BD.
Set oConn = Server.CreateObject ("ADODB.Connection")
Set RS = Server.CreateObject ("ADODB.RecordSet")
oConn.Open "DRIVER={Microsoft Access Driver (*.mdb)}; DBQ=" & Server.MapPath("\usuario\db\basedatos.mdb")
'Seleccionamos de la tabla solo los registros que concuerden con el usuario del Form.
sql = "SELECT * FROM prueba WHERE usuario='"&usuario&"'"
RS.Open sql, oConn, 1, 1
'Si el valor EOF (fin de la tabla elegida) es verdadero, no existe el usuario.
If RS.EOF = True then
Response.Write "<BR>Ese usuario no existe"
'Si el campo de la tabla es igual a nuestra variable, estas logueado.
ElseIf RS.Fields("password") = password then
Response.Write "<BR>Te logueaste con exito... Bienvenido <B>"& usuario &"</B>"
'Otra cosa, seria cuando el password no es la que seleccionamos.
Else
Response.Write "<BR>Esa contraseña no concuerda con el usuario ingresado"
End If
'Limpiamos y cerramos.
RS.Close
oConn.Close
Set RS = Nothing
Set oConn = Nothing
End If
%>


Aqui abajo esta el código funcionando para que lo prueben y entiendan mejor. La tabla que estoy utilizando, la creé en Access, y es la siguiente:

Completa el Formulario 

Principio del formulario

Usuario: [image: image4.wmf]


Contraseña: [image: image5.wmf]

Final del formulario

